

Pendahuluan

Telepon adalah salah satu teknologi komunikasi yang sangat dibutuhkan saat ini. Hal ini terbukti pada banyaknya jumlah pengguna telepon saat ini, baik telepon rumah ataupun telepon selular.

Di satu sisi, kehadiran telepon sangatlah dibutuhkan. Hampir di setiap rumah di DKI Jakarta dapat kita jumpai salah satu teknologi komunikasi ini. Namun di sisi lain ada hal yang menjadi dilema bagi masyarakat yaitu mahalnya tarif telepon saat ini. Sehingga para pengguna jasa telepon yang merasa sangat terbebani oleh kenaikan tarif tersebut dituntut untuk pandai-pandai menghemat pemakaian pulsa telepon.

Berdasarkan Pengumuman Menteri Perhubungan Nomor PM. 2 Tahun 2004 tanggal 30 Maret 2004 Tentang Pelaksanaan Restrukturisasi Sektor Telekomunikasi, PT. TELKOM memberlakukan penyesuaian tarif jasa telepon mulai tanggal 1 April 2004. Dengan prosentase kenaikan untuk tarif lokal dengan tarif percakapan per pulsa sebesar 28.21 persen dan

19.61 sampai 20.29 persen untuk biaya percakapan SLJJ dengan tarif percakapan per menit. (*Kep. Menhub No. PM.2 Tahun 2004*)

Bagi keluarga yang memiliki telepon rumah terkadang menyikapi tarif telepon yang mahal dengan menyegel pesawat telepon dengan gembok agar anggota keluarga tidak seenaknya menggunakan telepon. Hal tersebut biasanya dilakukan oleh keluarga yang khawatir terhadap tagihan telepon bulanan yang tinggi dikarenakan tidak terkontrolnya penggunaan telepon rumah oleh sekian banyak anggota keluarga. Bahkan terkadang pembantu rumah tangga pun ikut menggunakan telepon rumah dengan seenaknya tanpa sepengetahuan anggota keluarga pemilik rumah tersebut.

Sebenarnya penyegelan pesawat telepon rumah tidak perlu dilakukan pada keluarga manapun yang khawatir akan tagihan telepon bulanan yang tinggi jika kita dapat memantau penggunaan pulsa telepon dengan mudah. Dengan mudahnya kita mengetahui pemakaian telepon dalam nilai rupiah, kita dapat mengantisipasi pemakaian telepon yang berlebih sehingga kita dapat

membatasi pemakaian telepon per bulannya. Hadirnya mikrokontroler yang cukup marak penggunaannya belakangan ini seperti mikrokontroler *AT89C51* dapat menjadi sebuah solusi permasalahan tersebut. Untuk itu diupayakan membuat sebuah alat yang dapat menampilkan pemakaian telepon dalam nilai rupiah yang berbasis mikrokontroler *AT89C51*.

Buku ini akan memberikan panduan kepada pembaca seputar prinsip kerja dan langkah-langkah pembuatan alat tersebut. Layaknya sebuah tutorial, panduan yang terdapat pada buku ini dibuat sedemikian rupa sehingga memudahkan bagi para pembaca untuk memahaminya.


Alat Penampil Biaya Pemakaian Telepon Rumah Berbasis Mikrokontroler

Alat Penampil Biaya Pemakaian Telepon Rumah Berbasis Mikrokontroler merupakan sebuah alat yang berfungsi untuk menampilkan biaya pemakaian telepon rumah secara akumulasi pada setiap bulan. Mekanisme kerja alat ini yaitu dengan cara menghitung pulsa dan durasi waktu setiap pemakaian telepon kemudian mengkonversinya ke dalam nilai rupiah sesuai dengan tarif dasar telepon yang berlaku. Nilai rupiah yang didapat kemudian ditampilkannya pada *LCD* yang terdapat pada alat tersebut.

Cara kerja alat ini berbeda dengan Alat Penampil biaya pemakaian telepon pada WARTEL (Warung Telekomunikasi). Alat penampil biaya pemakaian telepon pada WARTEL hanya menampilkan biaya pada setiap pemakaian telepon dan kemudian *reset* kembali. Cara kerja alat ini selain menampilkan biaya pemakaian

telepon pada setiap pemakaian telepon juga berfungsi mengakumulasi seluruh biaya pemakaian telepon yang telah dilakukan tersebut selama satu bulan. Jadi dengan alat ini selain kita bisa mengetahui biaya pada saat pemakaian telepon kita pun dapat mengetahui total biaya pemakaian yang telah dilakukan setiap saat dan kita pun dapat mengetahui total tagihan biaya pemakaian telepon selama satu bulan.

Gambaran cara kerja alat ini dapat dilihat pada gambar 1 di bawah ini.




Gambar 1. Alur Kerja Alat Penampil Biaya Pemakaian Telepon Rumah Berbasis Mikrokontroler

Alat ini memanfaatkan sinyal *DTMF* sebagai input utamanya. Alat ini juga menggunakan listrik PLN sebagai supply tegangan utamanya. Namun alat ini juga sudah dilengkapi dengan *auto-switch power supply* yang berfungsi mengalihkan supply tegangan yang pada awalnya menggunakan listrik PLN akan berpindah ke listrik DC pada saat listrik PLN padam. Dengan adanya *auto-switch power supply* ini kita tidak perlu khawatir memori yang terdapat pada *RAM* akan hilang pada saat listrik PLN padam.

Langkah-langkah Pembuatan

A. Skema Diagram Blok Alat

Langkah awal untuk melakukan pembuatan alat ini diawali dengan memahami Skema Diagram Blok alat. Gambar 2 menunjukkan skema diagram blok dari alat yang akan kita buat.



Gambar 2. Skema Diagram Blok Alat Penampil Pemakaian Telepon Rumah Berbasis Mikrokontroler

Gambar di atas menerangkan alur kerja antar *hardware* dan juga antar blok. Pada blok input terdapat

pesawat telepon, *DTMF decoder*, *Hook detector*, dan *SPM detector*. Blok input berfungsi memberikan masukan berupa sinyal dan data yang akan diolah oleh mikrokontroler. Pada alat ini terdapat 3 buah input yang menjadi masukan mikrokontroler, yaitu sinyal dari *hook detector*, *SPM detector*, dan data dari *DTMF decoder*. Sinyal dari *hook detector* berfungsi sebagai pemberi tanda bahwa gagang telepon diangkat atau tidak. Sinyal dari *SPM detector* berfungsi sebagai pemberi tanda bahwa sambungan telepon telah terkoneksi. Data dari *DTMF decoder* berfungsi sebagai pemberi data dari *keypad* yang ditekan.

Blok proses merupakan blok yang berfungsi untuk mengolah masukan dari input. Mikrokontroler merupakan inti dari blok ini. Mikrokontroler bertugas merespon dan mengolah semua masukan yang diterima dari blok input.


Sinyal yang diterima dari *hook detector* menjadi tanda bagi mikrokontroler untuk memulai untuk bekerja. Proses perhitungan biaya pemakaian telepon pada mikrokontroler akan dimulai jika sinyal input dari *SPM detector* telah diterima. Data dari *DTMF decoder* berupa nomor telepon yang ditekan pada *keypad*. Nomor telepon

akan diolah oleh mikrokontroler kemudian ditampilkan pada LCD.

Blok output hanya terdiri dari LCD yang berfungsi sebagai *Alat Penampil* hasil proses yang akan ditampilkan.

B. Skema Rangkaian

Tahap selanjutnya adalah pembuatan skema rangkaian. Dari diagram blok yang telah dibahas sebelumnya, dapatlah dibuat skema rangkaian sebagaimana ditunjukkan pada gambar 3.



Gambar 3. Skema Rangkaian Alat Penampil Biaya Pemakaian Telepon Rumah


Gambar 3 menunjukkan konfigurasi antar *hardware* pada Alat Penampil Biaya Pemakaian Telepon Rumah Berbasis Mikrokontroler. Skema rangkaian

tersebut cukup menggambarkan rangkaian yang akan dibuat. Dari gambar 3 dapatlah dibuat sebuah prakiraan kebutuhan komponen dan biaya yang harus dikeluarkan untuk membuat sebuah Alat Penampil Biaya Pemakaian Telepon Rumah Berbasis Mikrokontroler.

Untuk memudahkan dalam mendata kebutuhan komponen yang akan digunakan, kita akan bahas per bagian dari skema rangkaian.

Blok Input

Rangkaian DTMF Receiver




Gambar 4. Rangkaian *DTMF Receiver*

Untuk membuat sebuah *DTMF receiver*, dibutuhkan beberapa komponen sebagai berikut :

- *DTMF receiver* MT8870 1 buah

- Kristal 3,5 MHz 1 buah
- Kapasitor 100 nF 2 buah
- Resistor 3,3 K Ω 3 buah

Rangkaian Hook Detector




Gambar 5. Rangkaian *Hook Detector*

Komponen yang dibutuhkan untuk membuat sebuah *Hook detector* yaitu :

- Optocoupler H11AA1 1 buah
- Resistor 470 Ohm 1 buah


Rangkaian SPM Detector



Gambar 6. Rangkaian *SPM Detector*

Rangkaian *SPM detector* hanya memanfaatkan saklar *push-botton* untuk menghasilkan simulasi sinyal *SPM*.

Blok Proses




Gambar 7. Minimum Sistem AT89C51

Blok proses hanya terdiri dari minimum sistem mikrokontroler AT89C51 yang secara rinci memiliki beberapa komponen sebagai berikut :

- AT89C51 1 buah
- Kristal 12 MHz 1 buah
- Resistor array 10 K Ω 1 buah
- Elco 10 μ F 1 buah
- Kapasitor 30 pF 2 buah
- Resistor 10 K Ω 1 buah

Blok Output



Gambar 8. Rangkaian LCD

Gambar 8 menunjukkan sebuah rangkaian LCD yang berfungsi sebagai output. Rangkaian ini terdiri dari beberapa komponen sebagai berikut :


- LCD HD44780U 1 unit
- Resistor variabel 10 K Ω 1 buah

Sekarang kita dapat membuat *Hardware* Alat Penampil Biaya Pemakaian Telepon Rumah Berbasis Mikrokontroler.

Perangkat Tambahan

Secara ideal, Alat Penampil Biaya Pemakaian Telepon Rumah selalu berada dalam kondisi menyala. Hal ini diperlukan agar memori yang tersimpan di dalam RAM mikrokontroler senantiasa tersimpan. Karena itu diperlukan sebuah *auto-switch power supply* yang berfungsi sebagai pemindah sumber listrik dari PLN ke sumber listrik DC pada saat listrik PLN padam.

Gambar 9 merupakan sebuah rangkaian *auto-switch power supply* yang dapat anda buat sebagai perangkat tambahan.



Gambar 9. Rangkaian *Auto-switch Power Supply*

Komponen yang dibutuhkan untuk membuat sebuah *auto-switch power supply* adalah :

- Trafo CT 2A 1 buah
- Dioda 4 4 buah
- Relay 12 vdc 1 buah
- Elco 2200 μ F 2 buah
- Elco 100 μ F 1 buah
- LM7805 2 buah
- Resistor variabel 10 K Ω 1 buah

C. Memasukkan Program

Setelah *hardware* alat yang akan kita buat sudah selesai, maka langkah selanjutnya adalah memasukkan program ke dalam *chip* mikrokontroler AT89C51.

Langkah awal yang harus kita lakukan untuk memasukkan program ke dalam mikrokontroler adalah terlebih dahulu menyiapkan program yang akan kita masukkan ke dalam *chip* tersebut.

Pada buku ini, penulis telah menyertakan program lengkap yang dapat anda tulis kemudian anda masukkan ke dalam *chip* mikrokontroler AT89C51. *Source code* program lengkap untuk Alat Penampil Biaya Pemakain Telepon Rumah Berbasis Mikrokontroler dapat dilihat pada lampiran 1.

Program yang dilampirkan pada buku ini dibuat menggunakan bahasa assembler. Silahkan anda gunakan editor yang anda sukai untuk menuliskannya. Setelah anda selesai membuat program anda, lanjutkan dengan memasukkan program tersebut ke dalam *chip* yang sudah anda sediakan.

Kelebihan dan Kekurangan Alat

Kelebihan

- Alat ini mampu menghitung biaya pemakaian telepon pada setiap pemakaian.
- Alat ini juga dapat mengakumulasi jumlah antara biaya pemakaian yang satu dengan biaya berikutnya terus menerus selama 1 bulan.
- Software tanggal yang terdapat pada alat ini berlaku hingga tahun 2099.
- Dilengkapi *auto-switch power supply*.
- Dapat berfungsi sebagai perbandingan tagihan bulanan yang dikeluarkan oleh TELKOM.

Kekurangan

- Untuk dapat diimplementasikan penggunaannya membutuhkan kerja sama dengan TELKOM mengenai akurasi tarif secara detail.
- *SPM detector* pada alat ini masih bersifat manual.
- Ukuran alat masih terlalu besar.

DAFTAR PUSTAKA

<http://www.atmel.com>

<http://www.cmlmicro.com>

<http://www.maxim-ic.com/errata>

<http://www.nsc.com>

<http://www.ustr.net>

<http://www.zarlink.com>

<http://www.8052.com>

H1601A LCD Module Manual, _____, _____.

Microcontroller Handbook, _____ : Intel Corporation, 1985.

Microcontroller Training, Jakarta : Laboratorium Elektronika FTUI & PS- E21, 2003.

Penyesuaian Tarif Jasa Telepon 2004, Jakarta : PT. Telekomunikasi Indonesia, Tbk, 2004.

Agfianto Eko Putra, Belajar Mikrokontroler, Yogyakarta : Penerbit Gava Media, 2002.

Paulus Andi Nalwan, Panduan Praktis Teknik Antarmuka dan Pemrograman Mikrokontroler AT89C51, Jakarta : PT. Elex Media Komputindo, 2003.

Sri Mulyono, Alat Pengaman Data Telepon Pada Warung Telekomunikasi (WARTEL) Ketika Aliran Listrik Terputus, Jakarta : Skripsi, 2002.

Suhana, Buku Pegangan Teknik Telekomunikasi, Jakarta : PT Pradnya Paramita, 1987.

Suryadi, Pengenalan Komputer, Jakarta : Penerbit Gunadarma, 1993.

Widjono HS, Bahasa Indonesia, Jakarta : Penerbit Jurusan MKU UNJ, 2003.

Lampiran 1

Listing Program Alat Penampil Biaya Pemakaian

Telepon Rumah Berbasis Mikrokontroler

```
;=====
; Inisialisasi awal
;=====

dv                bit        p3.4
cekhook          bit        p3.5
spm              bit        p3.6
data_lcd         equ        p0
rs               bit        p2.1
e                bit        p2.7
Dtmf             Equ        P3
total_tagihan    equ        20h
harga_awal       equ        40h
nomorawl         equ        52h
jmlh_sementara   equ        60h
data_calendar    equ        53h

;=====
;Main Program
;=====

                org        00h
                ljmp       mulai
                org        0bh
                ljmp       calendar

mulai:         call       init_lcd
                mov        a,#0ch
                call       command
                clr        a
```

```

        mov     ie,a
        setb   rs0
        setb   rs1
        mov    r0,#53h
bersih_53:
        mov    @r0,#0
        inc    r0
        cjne   r0,#59h,bersih_53

        mov    r3,#0
        jb     cekhook,$
set_calendar:
        mov    a,#0ffh
        jnb   dv,$
        mov    a,dtmf
        anl   a,#0fh
        cjne   a,#0ah,set_apa
        mov    a,#0h
set_apa:
        inc    r3
        cjne   r3,#1,sett_bln
        jmp    set_thn
sett_bln:
        cjne   r3,#2,set_tgl
        jmp    set_bln
set_thn:
        mov    r7,a
        jmp    tunggu_dv
set_bln:
        mov    r6,a
        jmp    tunggu_dv
set_tgl:
        mov    r5,a
        jmp    tgl_fix
tunggu_dv:
        jb     dv,$
        jmp    set_calendar

```

```

tgl_fix:
    mov     58h,r7
    mov     56h,r6
    mov     54h,r5
    mov     r4,#00h
    mov     r3,#00h
    mov     r2,#00h
    mov     r1,#00h
    mov     tmod,#11h
    mov     th0,#high (-50000)
    mov     tl0,#low (-50000)
    setb   tr0
    mov     ie,#82h
    clr     rs0
    clr     rs1
    call    lompatabu

AWAL:
    call    display_clock
    call    lompat
    mov     dptr,#pesanlcd1
    call    kirimpesan_lcd1
    jnb    cekhook,jmphook1
    call    display_clock
    MOV     DPTR,#PESANLCD
    call    kirimpesan_LCD
    mov     dptr,#pesanlcd1
    call    kirimpesan_lcd1
    jnb    cekhook,jmphook1
    ajmp   awal
jmphook1: jmp hook1
LOMPAT:
    mov     a,#083h
    call    command
    mov     a,#'R'
    call    char
    mov     a,#'p'
    call    char

```

```

        mov     r0,#44h
sublomp:
        mov     a,@r0
        add     a,#30h
        call    cha
        dec     r0
        cjne   r0,#3fh,sublomp
        ret
lompatabu:
        mov     r0,#harga_awal
bersih_r0:
        mov     @r0,#0
        inc     r0
        mov     a,r0
        cjne   a,#49h,bersih_r0
        mov     r0,#harga_awal
        mov     @r0,#0
        inc     r0
        mov     @r0,#0
        inc     r0
        mov     @r0,#1
        inc     r0
        mov     @r0,#4
        inc     r0
        mov     @r0,#3
        ret
lompat1:
AWAL1:
        call    display_clock
        call    display_harga
        mov     dptr,#pesanlcd1
        call    kirimpesan_lcd1
        jnb    cekhook,hook1
        call    display_clock
        MOV     DPTR,#PESANLCD
        call    kirimpesan_LCD
        mov     dptr,#pesanlcd1
        call    kirimpesan_lcd1

```

```

                jnb      cekhook, hook1
                ajmp     awal1

;=====
;Welcome Note Program
;=====

KIRIMPESAN_LCD:
    mov        a, #080h
    call       command
LOOPKIRIMPESAN_LCD:
    MOV        A, #00H
    MOVC       A, @A+DPTR
    cjne       a, #0fh, KIRIM_LCD
    RET
KIRIM_LCD:
    LCALL      CHAR
    INC        DPTR
    AJMP       LOOPKIRIMPESAN_LCD
pesanlcd:
    db         '"SILAHKAN TEKAN"', 0fh
KIRIMPESAN_LCD1:
    mov        a, #080h
    call       command
LOOPKIRIMPESAN_LCD1:
    MOV        A, #00H
    MOVC       A, @A+DPTR
    cjne       a, #0fh, KIRIM_LCD1
    RET
KIRIM_LCD1:
    LCALL      CHAR
    INC        DPTR
    AJMP       LOOPKIRIMPESAN_LCD1
pesanlcd1:
    db         '                ', 0fh

;=====

```

```

;Checking DTMF Program
;=====

hook1:  jb      cekhook,selesai
        mov     a,#01h
        call   command
        mov     r0,#15
hook:    clr     a
        jb     cekhook,selesai
        mov     72h,#0
Check_DTMF:
        Mov    A,#0FFH
        jb    spm,jmp_awalhitung
        Jnb   dv,Back_Check_DTMF
        Mov   A,Dtmf
        Anl   A,#0FH
        Cjne  A,#0AH,Check_DTMF_1
        Mov   A,#0h
Check_DTMF_1:
        mov   51h,a
        add   a,#30h
        call  char
        dec   r0
        cjne r0,#14,tahan
        mov   a,51h
        mov   nomorawl,a
tahan:   jb     cekhook,selesai
        Jb    dv,tahan
Back_Check_DTMF:
        jmp   hook
jmp_awalhitung:
        jmp   awalhitung

;=====
;Data Storage Program

```

```

;=====

scan:   mov     a,@r0
        add     a,@r1
        add     a,72h
        da      a
        mov     72h,#0
        jb     acc.4,isi
        mov     @r1,a
        inc    r0
        inc    r1
        cjne   r1,#49h,scan
        ljmp   lompat1
selesai:
        mov     a,#01h
        call   command
        mov     r1,#harga_awal
        mov     72h,#0
        mov     r0,#jmlh_sementara
        ljmp   scan
isi:    mov     72h,#1
        clr    acc.4
        mov     @r1,a
        inc    r0
        inc    r1
        cjne   r1,#49h,scan
        ljmp   lompat1
display_harga:
        mov     a,#83h
        call   command
        mov     a,#'R'
        call   char
        mov     a,#'p'
        call   char
        mov     r0,#48h
total1: mov     a,@r0
        cjne   a,#0,notzero

```

```

                dec     r0
                cjne   r0,#3fh,total1
                ret
total2: mov     a,@r0
notzero: add    a,#30h
                lcall  char
                dec    r0
                cjne   r0,#3fh,total2
                ret

```

```

;=====
;Pulse Counter Program
;=====

```

awalhitung:

```

    mov     a,#01h
    call   command
    mov     a,nomorawl
    cjne   a,#0,hitlok
    jmp    hitsljj

```

hitlok:

```

    mov     r0,#jmlh_sementara
    mov     r1,#8
    ljmp   bersihkan

```

hitsljj:

```

    mov     r0,#jmlh_sementara
    mov     r1,#8

```

bersihkan1:

```

    mov     @r0,#0
    inc     r0
    djnz   r1,bersihkan1

```

hitungsljj: ljmp hitsljjmulai

hitungsljj1: jb cekhook,jmpselesai

```

    call   menit
    sjmp   hitsljjmulai

```

jmpselesai: jmp selesai

hitunglokal: ljmp hitlokmulai

```

hitunglokal1:  jb cekhook,jmpselesai
                call    menit
                call    menit
                call    menit
                sjmp    hitlokmulai
hitsljmulai:
                mov     r0,#30h
                mov     71h,#163
                sjmp    start
bersihkan:
                mov     @r0,#0
                inc     r0
                djnz   r1,bersihkan
                ljmp   hitunglokal
hitlokmulai:
                mov     r0,#30h
                mov     71h,#250
start:
                mov     a,71h
                mov     70h,#0
                mov     r2,#0
                mov     r3,#100
                clr     c
ratusan:
                subb   a,r3
                jnc   hitung
                clr     c
                mov     b,r2
                mov     @r0,b
                inc     r0
                mov     r2,#0
                add    a,r3
                mov     r3,#10
                clr     c
puluhan:
                subb   a,r3
                jnc   hitung1

```

```

        clr     c
        mov     b,r2
        mov     @r0,b
        inc     r0
        add     a,r3
        mov     @r0,a
        acall   jumlah
        mov     a,71h
        cjne    a,#250,hitungsljj2
hitunglokal2:
        ljmp    hitunglokal1
hitungsljj2:
        ljmp    hitungsljj1
hitung:
        inc     r2
        sjmp    ratusan
hitung1:
        inc     r2
        sjmp    puluhan

jumlah:
        clr     c
        mov     r1,#jmlh_sementara
        add     a,@r1
        da     a
        jb     acc.4,naik
        mov     @r1,a
isipuluhan:
        dec     r0
        inc     r1
        clr     c
        mov     a,@r0
        add     a,@r1
        add     a,70h
        da     a
        mov     70h,#0
        jb     acc.4,naik1

```

```

        mov     @r1,a
isiratusan:
        dec     r0
        inc     r1
        clr     c
        mov     a,@r0
        add     a,@r1
        add     a,70h
        da     a
        mov     70h,#0
        jb     acc.4,naik2
        mov     @r1,a
isiribuan:
        mov     a,#0
        inc     r1
        clr     c
        add     a,@r1
        add     a,70h
        da     a
        mov     70h,#0
        jb     acc.4,naik3
        mov     @r1,a
isi10ribu:
        mov     a,#0
        inc     r1
        clr     c
        add     a,@r1
        add     a,70h
        da     a
        mov     70h,#0
        jb     acc.4,naik4
        mov     @r1,a
isi100ribu:
        mov     a,#0
        inc     r1
        clr     c
        add     a,@r1

```

```

        add    a,70h
        da     a
        mov    70h,#0
        jb     acc.4,naik5
        mov    @r1,a
isi1jt:
        mov    a,#0
        inc    r1
        clr    c
        add    a,@r1
        add    a,70h
        da     a
        mov    70h,#0
        jb     acc.4,naik6
        mov    @r1,a
isi10jt:
        mov    a,#0
        inc    r1
        clr    c
        add    a,@r1
        add    a,70h
        da     a
        mov    70h,#0
        mov    @r1,a
        call   kirim_rupiah
        ret
naik:
        mov    70h,#1
        clr    acc.4
        mov    @r1,a
        sjmp   isipuluhan
naik1:
        mov    70h,#1
        clr    acc.4
        mov    @r1,a
        sjmp   isiratusan
naik2:

```

```

        mov     70h,#1
        clr     acc.4
        mov     @r1,a
        sjmp    isiribuan
naik3:
        mov     70h,#1
        clr     acc.4
        mov     @r1,a
        sjmp    isi10ribu
naik4:
        mov     70h,#1
        clr     acc.4
        mov     @r1,a
        sjmp    isi100ribu
naik5:
        mov     70h,#1
        clr     acc.4
        mov     @r1,a
        sjmp    isi1jt
naik6:
        mov     70h,#1
        clr     acc.4
        mov     @r1,a
        sjmp    isi10jt

```

```

;=====
;Sending Rupiah Program
;=====

```

```

kirim_rupiah:
        mov     r2,#85h
        mov     r1,#68h
balik:  mov     a,@r1
        cjne    a,#0,notclear
        dec     r1
        inc     r2
        cjne    r1,#5fh,balik

```

```

        ret
notclear:
        mov     a,r2
        call   command
        mov     a,#'R'
        call   char
        mov     a,#'p'
        call   char
balik1:  mov     a,@r1
        add     a,#30h
        lcall  char
        dec     r1
        cjne   r1,#5fh,balik1
        ret

```

```

;=====
;LCD Initialization Program
;=====

```

```

init_lcd:
        MOV     A,#38H
        call   command
        MOV     A,#01H
        call   command
        MOV     A,#0EH
        call   command
        MOV     A,#06H
        call   command
        RET
command:
        clr     rs
        mov     data_lcd,a
        setb   e
        call   delay
        clr     e
        ret
char:   setb   rs
        mov     data_lcd,a

```

```

        setb     e
        call    delay
        clr     e
        ret
delay:  mov     r7,#0
dla:   mov     r6,#0
       djnz   r6,$
       djnz   r7,dla
       ret

```

```

;=====
;One Minute Delay Program
;=====

```

```

menit:
       mov     r4,#60
detik:  mov     r5,#20
repeat: mov     th1,#high (-50000)
       mov     tl1,#low (-50000)
       setb   tr1
wait:   jb     cekhook,lompatselesai
       jnb   tf1,wait
       clr   tf1
       clr   tr1
       djnz  r5,repeat
       djnz  r4,detik
       ret
lompatselesai:
       ljmp   selesai

```

```

;=====
;Clendar program
;=====

```

```

calendar:
    setb    rs0
    setb    rs1
    inc     r1
    cjne   r1,#20h,out1
    inc     r2
    cjne   r2,#60h,out2
    inc     r3
    cjne   r3,#60h,out3
    inc     r4
    cjne   r4,#24h,out4
    call   tanggal
    jmp    hari
bln1:    mov     r6,#1
tgl1:    call   lompatabu
         mov     r5,#1
         call   isitgl
hari:    mov     r4,#00h
out4:    mov     r3,#00h
out3:    mov     r2,#00h
out2:    mov     r1,#00h
out1:    mov     th0,#high (-50000)
         mov     tl0,#low (-50000)
         clr     rs0
         clr     rs1
         reti

tanggal:
    inc     r5
    mov     a,r6
    cjne   a,#13,blm_akhir_thn
    inc     r7
    jmp    bln1
blm_akhir_thn:
    mov     a,r7
    mov     b,#4

```

```

        div     ab
        mov     a,b
        jz      kabisat
bulan:
        mov     a,r6
        cjne   a,#2,notfeb
        mov     a,r5
        cjne   a,#29,isitgl
        inc    r6
        jmp    tgl1

notfeb: mov     a,r6
        cjne   a,#1,bkn_januari
        jmp    bln_31
bkn_januari:
        cjne   a,#3,bkn_maret
        jmp    bln_31
bkn_maret:
        cjne   a,#4,bkn_april
        jmp    bln_30
bkn_april:
        cjne   a,#5,bkn_mei
        jmp    bln_31
bkn_mei:
        cjne   a,#6,bkn_juni
        jmp    bln_30
bkn_juni:
        cjne   a,#7,bkn_juli
        jmp    bln_31
bkn_juli:
        cjne   a,#8,bkn_agustus
        jmp    bln_31
bkn_agustus:
        cjne   a,#9,bkn_september
        jmp    bln_30
bkn_september:
        cjne   a,#10,bkn_oktober

```

```

                jmp        bln_31
bkn_oktober:   cjne      a,#11,bln_31
                jmp        bln_30
bln_31:        mov       a,r5
                cjne      a,#32,isitgl
                inc       r6
                jmp       tgl1

bln_30:        mov       a,r5
                cjne      a,#31,isitgl
                inc       r6
                jmp       tgl1

kabisat:      mov       a,r6
                cjne      a,#2,notfeb
                mov       a,r5
                cjne      a,#30,isitgl
                inc       r6
                jmp       tgl1

isitgl:       clr        a
                clr        c
                mov       r0,#-1
                mov       a,r5

loop1:        mov       b,#10
                subb      a,b
                inc       r0
                jnc       loop1
                add       a,#10
                mov       54h,a
                clr        c
                mov       53h,r0

```

```

        mov     r0,#-1
        mov     a,r6
loop2:
        mov     b,#10
        subb    a,b
        inc     r0
        jnc     loop2
        add     a,#10
        mov     56h,a
        clr     c
        mov     55h,r0
        mov     r0,#-1
        mov     a,r7
loop3:
        mov     b,#10
        subb    a,b
        inc     r0
        jnc     loop3
        add     a,#10
        mov     58h,a
        clr     c
        mov     57h,r0
        jmp     hari
display_clock:
        mov     a,#0c3h
        call    command
        mov     a,53h

        add     a,#30h
        call    char
        mov     a,54h
        add     a,#30h
        call    char
        mov     a,#'-'
        call    char
        mov     a,55h

```

```
add    a,#30h
call   char
mov    a,56h
add    a,#30h
call   char
mov    a,#'-'
call   char
mov    a,#'2'
call   char
mov    a,#'0'
call   char
mov    a,57h
add    a,#30h
call   char
mov    a,58h
add    a,#30h
call   char
ret
end
```

Spesifikasi dan Petunjuk Penggunaan Alat Penampil Biaya Pemakaian Telepon Rumah Berbasis Mikrokontroler

Spesifikasi Alat :

Sumber Tegangan

AC	:	220 V
DC	:	12 V
Battery	:	Alkaline 23AE; 12V
Konsumsi Daya	:	456 mW

Petunjuk Penggunaan Alat :

1. Pastikan saklar power dalam keadaan *off*.
2. Hubungkan kabel power pada alat ke sumber tegangan AC
3. Pasang baterai pada alat
4. Hidupkan alat dengan memosisikan saklar pada posisi *on*
5. Lakukan pengisian tanggal dengan cara menekan 3 nomor setelah alat dihidupkan. Nomor pertama menunjukkan bilangan tahun, nomor kedua

menunjukkan bilangan bulan, dan ketiga adalah tanggal

6. Setelah prosedur di atas selesai dilakukan, maka alat sudah siap digunakan

Keterangan :

Alat ini akan menampilkan biaya setiap pemakaian telepon dilakukan, kemudian biaya pemakaian tersebut akan dijumlah dengan biaya sebelumnya. Proses tersebut berlangsung hingga 1 bulan.

Perhatian !!!

Jangan pernah mematikan alat, karena data yang ada akan hilang.

Biografi dan Profil



Khoirul Umam. Lahir di Jakarta, 10 Desember 1980. Menamatkan Madrasah di MAN 2 Jakarta Timur pada tahun 1998. Menyelesaikan program S1 pada jurusan Teknik Elektro Program Studi Elektronika di

Universitas Negeri Jakarta pada tahun 2005. Saat ini bekerja di PT Nurul Fikri Cipta Inovasi sebagai Staff Teknik Komputer. Kompetensi inti adalah pada bidang Microcontroller dan Robotic.

Berpengalaman sebagai Konsultan Tugas Akhir Mahasiswa Elektronika Dan Teknik Komputer dari berbagai kampus dan juga perusahaan dalam pembuatan alat-alat otomatis berbasis microcontroller. Berpengalaman sebagai Trainer Indoor, Outdoor, dan Outbound Training, pembicara materi-materi organisasi dan life skill di berbagai lembaga, sekolah, dan kampus.

Aktif dalam berbagai organisasi pelajar, kemahasiswaan, sosial, dan politik. Ketua Forum Alumni Rohis MAN 2 dan MTsN 7 (FARM2) Jakarta, Ketua Divisi PSDM Partai Keadilan Sejahtera Dewan Pimpinan Ranting Kelurahan Pasir Gunung Selatan Cimanggis-Depok.

Khoirul Umam termasuk pendiri Kelompok Studi Dan Implementasi Ilmu Pengetahuan Dan Teknologi (KSI-IPTEK), Sentra Kajian Informasi Dan Literatur (SKIL) Learning Center.

Informasi lebih lanjut tentang penulis ini bisa didapat melalui:

Email: ksi_iptek@yahoo.co.id